# 2d Beam structure – solution by MATHEMATICA and ANSYS

**PROBLEM 7.1**




**Finite element model of a welded 2D beam structure.**

*The beam structure above consists of linear elastic beam elements (tension-compression and bending).*

a. *Calculate the nodal displacements and the reactions by the finite element method using MATHEMATICA. Animate the deformed shape of the structure. Calculate the beam diagram data (normal force, shear force, bending moment) of the elements.*
b. *Solve the problem by ANSYS, verify the MATHEMATICA solution*

*Data:* $a = 1$ m, $b = 1.2$ m, $c = 1.1$ m

$A = 150 \cdot 10^{-6}$ m$^2$, $I_z = 21 \cdot 10^{-8}$ m$^4$, $E = 200$ GPa

$p_1 = 600$ N/m, $p_2 = 800$ N/m, $F = 10\,000$ N

## MATHEMATICA solution

**Definition of the Hermitian polynomials** *(use copy & paste)*

```
N1[xi_]=1/4*(2-3*xi+xi^3);
N2[xi_]=L/8*(1-xi-xi^2+xi^3);
N3[xi_]=1/4*(2+3*xi-xi^3);
N4[xi_]=-L/8*(1+xi-xi^2-xi^3);
```

**Vector of interpolation functions, bending**

```
Nbend={{N1[xi]},{N2[xi]},{N3[xi]},{N4[xi]}};
Bbend=(4/L^2)*D[Nbend,{xi,2}];
Bbendt=Transpose[Bbend];
```

**Construction of bending stiffness matrix**

```
Kebend=Iz*Ex*L/2*Integrate[Bbend.Bbendt,{xi,-1,1}];
```

**Definition of interpolation functions for tension-compression**

```
N5[xi_]=1/2*(1-xi);
N6[xi_]=1/2*(1+xi);
```

**Vector of interpolation functions, tension-compression**

```
Ntens={{N5[xi]},{N6[xi]}};
Ntenst=Transpose[Ntens];
Btens=(2/L)*D[Ntens,xi];
Btenst=Transpose[Btens];
```

**Construction of extensional stiffness matrix**

```
Ketens=A*Ex*L/2*Integrate[Btens.Btenst,{xi,-1,1}];
```

**Construction of the stiffness matrix, general beam element (bending+tension/compr.)**

*(use copy & paste)*

```
Ke=Table[0,{6},{6}];
Ke[[{1,4},{1,4}]]=Ketens;
Ke[[{2,3,5,6},{2,3,5,6}]]=Kebend;
```

**Construction of the element stiffness matrices in the local (element) C.S.**

```
Ke1loc=Ke/.{L->L1};
Ke2loc= Ke/.{L->L2};
```

**Definition of the transformation matrices** *(use copy & paste)*

```
T1={{ c1,-s1,0,0,0,0},{ s1,c1,0,0,0,0},{0,0,1,0,0,0},
{0,0,0,c1,-s1,0},{0,0,0,s1,c1,0},{0,0,0,0,0,1}};
T1t=Transpose[T1];
T2={{ c2,-s2,0,0,0,0},{ s2,c2,0,0,0,0},{0,0,1,0,0,0},
{0,0,0,c2,-s2,0},{0,0,0,s2,c2,0},{0,0,0,0,0,1}};
T2t=Transpose[T2];
```

**Check the orthogonality of the transformation matrices**

```
T1.T1t;T2.T2t;
```

**Transformation of the stiffness matrices to the global C.S.**

```
Ke1=T1.Ke1loc.T1t;
Ke2=T2.Ke2loc.T2t;
```

**Construction of the structural level matrices by expanding the element matrices**

```
K=Table[0,{9},{9}];
K[[{1,2,3,4,5,6},{1,2,3,4,5,6}]]= K[[{1,2,3,4,5,6},{1,2,3,4,5,6}]]+Ke1;
K[[{4,5,6,7,8,9},{4,5,6,7,8,9}]]= K[[{4,5,6,7,8,9},{4,5,6,7,8,9}]]+Ke2;
```

**Definition of the local line load function and the load vector for element 1**

```
p[xi_]=-(p1+p2)/2+(p1-p2)/2*xi;
F1ploc=(L1/2)*Integrate[{0,p[xi]*N1[xi], p[xi]*N2[xi],0, p[xi]*N3[xi], p[xi]*N4[xi]},{xi,-
1,1}]/.{L->L1}
```

**Transformation of the load vector to the global C.S.**

```
F1p=T1.F1ploc;
```

**Construction of the structural force and nodal displacement vector**

```
Ay=.;M1=.;Bx=.;By=.;u1=.;u2=.;v2=.;fi2=.;fi3=.;
```

```
F={0,Ay,M1,0,-Fc,0,Bx,By,0}+Flatten[{F1p,{0,0,0}}];
U={u1,0,0,u2,v2,fi2,0,0,fi3};
```

**Definition of the geometrical and material properties (units are in SI)**

```
a=1;b=1.2;c=1.1;L1=Sqrt[a^2+c^2];L2=b;Iz=21*10^-8;
Ex=200*10^9;A=150*10^-6;p1=600;p2=800;Fc=10000;
```

**Definition of nodal coordinates, plot of the structure**

```
x1=0;y1=0;x2=a;y2=c;x3=a+b;y3=c;
nodes={{x1,y1},{x2,y2},{x3,y3}};
Graphics[{
{Line[nodes]}
}]
```

**Definition of the transformation matrix components**

```
c1=(x2-x1)/L1;s1=(y2-y1)/L1;c2=(x3-x2)/L2;s2=(y3-y2)/L2;
```

**Construction and solution of the structural equation system, nodal displacements**

```
adof={1, 4,5,6,9};
Kcon=K[[adof,adof]];
Fcon=F[[adof]];
Ucon=LinearSolve[Kcon,Fcon]
```

*Results:*
$u_1$ = -0.0283 m, $u_2$ = $0.308\cdot 10^{-4}$ m, $v_2$ = -0.02619 m, $\varphi_2$ = -0.009075 rad, $\varphi_3$ = -0.0372 rad,

**Calculation and assignment of the reaction**

```
U[[adof]]=Ucon;
fixdof={2,3,7,8};
sol=Solve[(K.U)[[fixdof]]==F[[fixdof]],{Ay,M1,Bx,By}];
{Ay,M1,Bx,By}={Ay,M1,Bx,By}/.sol[[1]];
```

*Results:* $A_y$ = 7996.26 N, $B_x$ = -770 N, $B_y$ = 2703.73 N, $M_A$ = 4015.11 Nm

**Calculation of the element displacement vectors in the global and local C.S.s**

```
ue1=U[[{1,2,3,4,5,6}]];
ue2= U[[{4,5,6,7,8,9}]];
ue1loc=T1t.ue1;ue2loc=T2t.ue2;
```

**Calculation of the internal force vectors (=beam diagram data), local C.S.**

```
Fe1intloc=Ke1loc.ue1loc -F1ploc;
Fe2intloc=Ke2loc.ue2loc;
```

| | | | | |
|---|---|---|---|---|
| *Results:* | *Element 1* | $N_1$ = 5916.76 N, | $V_1$ = 5378.87 N, | $M_1$ = 4015.11 Nm, |
| | | $N_2$ = -5916.76 N, | $V_2$ = -4338.24 N, | $M_2$ = 3244.48 Nm |
| | *Element 2* | $N_2$ = 770 N, | $V_2$ = -2703.74 N, | $M_2$ = -3244.48 Nm, |
| | | $N_3$ = -770 N, | $V_3$ = 2703.74, | $M_3$ = 0 |

# ANSYS Solution

**Analysis type**
PREFERENCES – STRUCTURAL
File menu / Change Title / Title: "*2D BEAM structure*", refresh the screen by the mouse

**Element type – BEAM2D**
PREPROCESSOR / ELEMENT TYPES / ADD/EDIT/DELETE / ADD / BEAM / 2D ELASTIC 3 / OK / CLOSE
/ REAL CONSTANTS / ADD/EDIT/DELETE / ADD / OK / AREA=150E-6, IZZ=21E-8, HEIGHT=0.01 / OK

**Material properties**
PREPROCESSOR / MATERIAL PROPERTIES / MATERIAL MODELS / STRUCTURAL / LINEAR / ELASTIC / ISOTROPIC / - EX=200e9, PRXY=0.3

Material menu / Exit

**Geometry of the problem**
PREPROCESSOR / MODELING / CREATE / KEYPOINTS / IN ACTICE CS / X-Y-Z COORDINATES / APPLY
$x$=0, $y$=0, $z$=0
$x$=1, $y$=1.1, $z$=0
$x$=2.2, $y$=1.1, $z$=0
/ LINES / LINES / STRAIGHT LINE – Create the lines using the mouse

**Meshing**
***Definition of the element numbers on all lines***
PREPROCESSOR / MESHING / SIZE CNTRLS / MANUALSIZE / LINES / PICKED LINES / PICK ALL / NO. OF ELEMENT DIVISIONS: 1

PREPROCESSOR / MESHING / MESH / LINES / PICK ALL

Plot menu / Multi-Plots – Plot all objects

**Kinematic constraints**
PREPROCESSOR / LOADS / DEFINE LOADS / APPLY / STUCTURAL / DISPLACEMENT / ON NODES / on node 1 – UY, ROTZ
on node 3 – UX, UY

**Define the external loads**
PREPROCESSOR / LOADS / DEFINE LOADS / APPLY / STUCTURAL / PRESSURE / ON BEAMS / on the first element: VALI = 600, VAL2 = 800

PREPROCESSOR / LOADS / DEFINE LOADS / APPLY / STUCTURAL / FORCE/MOMENT /
ON NODES / on node 2 – FY = - 10 000

**Solution**
SOLUTION / SOLVE / CURRENT LS

**Results - Postprocessing**

GENERAL POSTPROC / PLOT RESULTS / DEFORMED SHAPE
/ CONTOUR PLOT / NODAL SOLU
Plot Ctrls menu / Animate / Deformed Shape – animate the deformed shape
List menu / Results / Nodal Solution / DOF Solution / Disp. vector sum
Rot. vector sum

***Beam diagrams***

GENERAL POSTPROC / ELEMENT TABLE / DEFINE TABLE
By sequence num – LS, 1 or 4 - SDIR – axial (normal) stress
SMISC, 1 or 7 - MFORX - normal force
SMISC, 2 or 8 - MFORY - shear force
SMISC, 6 or 12 – MMOMZ – moment

***Display and list the beam diagrams/values***

GENERAL POSTPROC / PLOT RESULTS / CONTOUR PLOT / LINE ELEM RES –
choose the component
GENERAL POSTPROC / ELEMENT TABLE / LIST ELEM TABLE

*…..compare the results to the MATHEMATICA solution………*